

Mathematical Discoveries in BASIC

11

Richard J. Shumway
Alfinio Flores
1983

MATHEMATICAL DISCOVERIES IN BASIC

Mathematics
through
Programming
in BASIC

Richard Shumway
Alfinio Flores

Mathematical Discoveries in BASIC

by

Richard J. Shumway
Ohio State University

and

Alfinio Flores
CIMAT, Guanajuato

Chapters

1. Counting
2. Functions and Tables
3. Geometric Patterns, and Graphs
4. Decimals, Ratio, and Percent
5. Probability and Predicting
6. Approximation
7. Applications

INTRODUCTION

Mathematical Discoveries in BASIC is based on the result that students of all ages can learn significant, important mathematics from very short programs in BASIC. Each page begins with a short program which, nevertheless, illustrates important, big ideas in mathematics. None of the programs are silly, or just examples to show how commands work. All programs do real work. Using these short programs will allow you to give every student a successful starting point. Each program can be entered into any computer and will do interesting work when run. Subsequent questions and suggestions help students explore the mathematics involved and add lines to the program to create new mathematical programs. We believe such an approach allows students to learn good programming habits, develop their own programs, and learn mathematics appropriate for today and the future.

Students do not need prior experience with computers to use these materials. However, even students with a background in programming will find the materials appropriate because of the mathematics which the programs do and cause students to learn. Student programming develops the deeper sort of understanding of mathematics which only seems to come when one must teach the idea to someone else (the someone else being the computer in this case). Computer programming and learning mathematics can go hand-in-hand.

Content: We have chosen for the chapter headings the general topics of Counting, Functions and Tables, Geometric Patterns and Graphs, Decimals, Ratios, and Percent, Probability and Predicting, Approximation, and Applications. You will also find concepts such as place value, large numbers, powers of ten, scientific notation, exponential growth, least common multiples, prime numbers, divisibility, rounding, estimation, integers, equations, inequalities, fractions, and statistics are treated in a reasonable, natural way. Of course you can expect students to become computer literate in BASIC. In addition, these activities also encourage reading skills, attention to detail, following directions, expose students to unifying concepts such as variable, functions, number, cartesian coordinates, limits, and recursive definitions, and attempt to develop fundamental thinking skills such as generalization, modeling, logical thinking, and problem solving.

Classroom suggestions: Each activity begins with a basic program such as:

```
10 FOR N=1 TO 12
20 PRINT N,N+N
30 NEXT N
```

Ask students to discuss the program briefly and try to predict the output, or what they think the program will do. If you have just one computer, enter the program as a class, sharing responsibilities such as typing and proofreading. If you have several computers or one for each student, the materials may also be used individually or in small groups. We have tried to make enough suggestions and comments to allow students to work independently if you wish. For this program, students would be encouraged to make modifications to the program to obtain other suggested outputs. Even in cases where you have plenty of computers, class discussions centered around sharing programs, discoveries and mathematics, or encouraging students to propose other uses of the programs and the mathematics are very important and need your encouragement and stimulation.

It is our view a substantial change in our curriculum in mathematics can be made as young students have the opportunity to write programs of the type in this book. It seems to us we can use daily student programming as an integral part of all the mathematics we teach. The richness of mathematics encountered will be multiplied dramatically, more young students will see mathematics as a dynamic, creative activity, and teaching and learning mathematics will be a lively, active sport indeed. Good luck!

A Note On Machines: The programs have been carefully designed to work on almost all micro-computers. For example, Apple, Radio Shack, IBM-PC, Atari, Texas Instruments, Timex-Sinclair, Vic-20, PET, and many other computers will run all the non-graphics programs. In the case of the graphics programs, only the lines containing the special graphics commands unique to a specific computer need be altered. Students learn to make these slight modifications quickly. Each activity has been tested on all of the above computers.

Note: A portion of this introduction is reprinted from Shumway, R. J. Young Children, Programming and Mathematics, a chapter in Computers and Mathematics Education (ed. Viggo Hansen) Reston, VA: National Council of Teachers of Mathematics, 1984 Yearbook.

Note: For more examples of computers and young students see Computer Corner, a monthly column in the Arithmetic Teacher (ed. Richard Shumway), NCTM, 1983-4.

Graphing

1. Try:

```
10 FOR N=0 TO 39
20 PLOT N,N
30 NEXT N
```

(Note: Program works, as written, on Timex-Sinclair 1000. Add any special graphics commands your computer requires. For example,

Apple: add lines:

```
5 GR
6 COLOR=7
```

Radio Shack:

```
replace 20 by
20 SET (N,N)
```

For others, check for graphics commands in manual.

2. Now you have fixed up the program for your computer. RUN the program again.

Draw what you got here:

Did you get a diagonal?



N,N

3. Change line 20 just slightly and see if you can get the other diagonal. Try several possibilities and record what you get.


4. See if you can predict what your computer will do for each.

Use one color for your prediction and another color for what the computer did.

Prediction color:

Computer color:

Using numbers to make pictures is one of the big ideas of mathematics. You are doing great!

	
\emptyset, N	$5, N$
$2\emptyset, N$	$N, 2\emptyset$
$N, N/2$	$39-N, N$

5. Teach the computer to draw a big X in the middle of the screen.

*6. Teach the computer to draw a big N in the middle of the screen. See if you can make another N about half the size of the big one.

Mathematics:

Cartesian coordinates
Graphing
Variables
translation

Computers:

FOR-NEXT
PLOT

Growing Numbers

1. Take a sheet of paper and fold it in half. Now fold it in half again (without unfolding the first fold). Continue folding and fill out the table below up to 7 folds.
2. Try this:

```

10 LET G=1
20 PRINT G
30 LET G=G+G
40 GOTO 20
    
```

Number of folds	Number of Thicknesses
0	1
1	2
2	4
3	8
4	
5	
6	
7	

Type RUN. To stop the program press CTL, C or BREAK. RUN the program again. Fill in the rest of the table. (You may have to be fast to stop the computer in time. On the Apple it helps to add line 3: 3 SPEED=200)

3. Run the program again. How many thicknesses are there after 9 folds?
-

4. Before we talk more about this problem, we have to talk about what the computer is printing. Here's what our computer did: (Type CONT to get your machine to continue if it stops along the way).

<u>At First</u>	<u>Later</u>
1	
2	1.4073749E+14
4	2.8147498E+14
8	5.6294995E+14
16	1.1258999E+15
32	.
64	.
128	.
256	.
.	.
.	.
.	.

and finally it just quit with the number 8.5070592E+37. What was the first "funny number" or number in scientific notation your computer printed?

What was the last number it printed before it quit by itself? _____

We'll talk more about these numbers later. They are pretty interesting.

What would happen if you used LET G=2*G?

Prediction _____

Actual _____

What would happen if you used LET G=G*G?

Prediction _____

Actual _____

Suprised?!

Change line 10 to:

10 LET G=2

and try again. What did you get? _____

Mathematics: Variable

Exponential Growth

Recursive definition

(defining new G from old G, line 20)

Big Numbers

Decimals

Scientific Notation

Applications (bacteria growth, think of others)

Computers: Computer Overflow

LET

GOTO

BREAK or CTL, C

Counting and Tables

```
10 FOR N = 1 TO 12
20 PRINT N, N+N
30 NEXT N
```

1. Before you RUN the above program, predict the output here:

2. Type LIST (enter) to see the program and then RUN (enter) twice just to see what the computer does and how the output compares with your prediction. Retype line 20 as follows:

```
20 PRINT N, N + N + N
```

- Now type LIST and RUN. Fill out the table with your results. Type LIST and RUN again if you would like to. Now change line again by typing:

```
20 PRINT N, 5 * N
```

- Now type LIST and RUN. Fill in the table with your results. Do your results have anything to do with clocks? _____ Why? _____

N	N+N+N	5*N
1		
2		
3		
4		
5		
6		
7		
8		
9		
10		
11		
12		

3. So far we have tried N+N, N+N+N, and 5*N in line 20. Here is a list of several more to try. Try them and describe the results:

Function	Results
N+N	Counts by 2's
N+N+N	Counts by 3's
5*N	Counts by
9*N	
2*N-1	
N-N	
1.59*N	
N*N	
N*N*N	
3*N	
N/2	

Did you notice that $N+N+N$ and $3*N$ gave the same results? _____
 Does that make sense? _____ Why? _____

What would be a shorter way to write $N+N+N+N+N$? _____
 How about $2*N + 5*N$? _____ See if the computer agrees with you by trying the programs with each functions, first the long one and then the shorter one.

4. Suppose you were selling school T-shirts for \$6.59 a shirt. Write a computer program that would fill out the following table:

```
10 FOR N=1 TO 12
20
30 NEXT N
```

Number of Shirts	Cost
1	\$ 6.59
2	
3	
4	
5	
6	
7	
8	
9	
10	
11	
12	

5. Now to really put the computer to work, change line 10 to:

```
10 FOR N=1 TO 100
```

Try any function you want for line 20, and use 30 NEXT N for line 30. Type LIST and RUN. What happened? _____

Are you impressed? _____
Try some others. Describe what happened.

Mathematics: Counting
Multiples
Variables
Multiplication as repeated addition
Functions

Computers: FOR-NEXT
PRINT
RUN
LIST

3. Let's return to the horse race. Which horse has the highest probability to advance in each toss? _____ Which one has the highest probability to win the race? _____ Will the favorite horse always win? _____ Most of the time? _____
4. This program simulates the race

```

20 FOR K = 2 TO 12
30 PRINT AT 0, 2*K;K
40 NEXT K
110 DIM V(12)
130 LET S = INT(6*RND+1)+INT(6*RND+1)
140 LET V(S) = V(S)+1
150 PRINT AT V(S)+1, 2*S; V(S)
160 IF V(S) = 9 THEN STOP
170 GOTO 130

```

Apple users:

```

5 GR
6 COLOR = 13
110 DIM V(12)
130 LET S = INT(6*RND(1)+1) + INT(6*RND(1)+1)
140 LET V(S) = V(S)+1
150 PLOT 2*S, 2*V(S)
160 IF V(S) = 9 THEN PRINT "WINNER IS";S:STOP
170 GOTO 130

```

Run the program several times. Does the favorite always win? _____

Is there ever an unexpected winner sometimes? (a dark horse).

Further activities

5. a) Modify line 160 to make the race shorter. Would you expect the favorite to win more often? Less often? Run the program several times.
- b) Make the race real short, say 3 instead of 9. What would you expect now? Run the program several times.
- c) What would happen with a longer race? Modify the program and run it several times.

Mathematics:

Randomness
Probability distribution
Counting
independence

Computers:

DIM
FOR-NEXT
PLOT

The Mad Adder

1. Run:

```

10 INPUT N
20 LET S = 0
30 FOR I = 1 TO N
40 LET S = S + I
50 NEXT I
60 PRINT N, S
70 GOTO 10
    
```

2. Complete the following table

using the above program

(IF $N = 4$, then $S = 1+2+3+4$?)

N	S (sum)
4	10
10	55
12	78
50	
100	
200	
500	
1000	
10000	
40000	

3. Make an N box, an S box, and

an I box and trace how the

program works. Does it

add the way you and I do?

4. Check this by hand. Does $1+2+3+4+5+6$

$+7+8+9+10$ "really" equal 55?

5. If you are interested in financial matters,

look up "rule 78" and installment loans.

How is 78 related to installment loans?

6. Now, check the numbers in the table

and see if you can find a formula so that

the program can be shorter and faster.

```

10 INPUT N
20 PRINT N, _____
30 GOTO 10
    
```

You fill in the blank. If you want


some help from Gauss, read on.

Gauss

Rumor has it when Gauss (1777-1855) was a boy one of his teachers, as discipline for the whole class, asked the students to find the sum of the numbers from 1 to 100. Gauss simply wrote 5050 on his slate and tossed

it on the teacher's desk.* How did he do it? And with no computer!

Maybe his thinking went this way:



Each pair of numbers adds to
101 (1+100, 2+99, 3+98, ...).

There are fifty pairs of numbers,
so the answer must be:

$$50 \times 101$$

or

$$5050$$

Now, if you had N numbers, what
formula would Gauss use? Go back
to the Mad Adder and try it.

7. What if N is odd? Does the same formula work? Why?
8. Can you change the first program to do Mad Multiplying?

Mathematics:

Sums (series)
Patterns
Algorithms

Computers:

INPUT
FOR-NEXT
LET S = S+I

*Boyer, C. B. A History of Mathematics. New York: John Wiley & Sons, Inc.,
1968, p. 544.

Sometimes bigger is better: random sampling

1. The mean or average of the numbers from 0 to 99 is

$$\frac{0+1+2+ \dots +97+98+99}{100} = \frac{4950}{100} = 49.5$$

(remember the mad adder)

Let's take a random sample of size 5 of these numbers and see how well the sample mean estimates the real mean.

This program tells the computer to select five numbers between 0 and 99 at random, and compute the average of the five.

```
90 LET S = 0
100 FOR N = 1 TO 5
110 LET X = INT(100*RND)
120 PRINT X;",";
130 LET S = S+X
140 NEXT N
150 PRINT "AVERAGE ";S/5
```

Run the program. Is the sample average close to the real one? Run the program several times. Write down the averages you got _____.


2. Add these two lines to the program to run the program 20 times.

```
20 FOR M = 1 TO 20
200 NEXT M
```

This is something nice about computers, it is so easy to tell them to do the same thing over and over again.

Run the extended program. Are all the sample averages close to the real mean? Did you get any average that is far away from 49.5 (like 3.2 or 96.6?) Where do most averages fall? Fill the table and draw a bar graph.

range	frequency
25-34.9	
35-44.9	
45-54.9	
55-64.9	
65-74.9	
75-84.9	



3. We can easily modify the program to select samples of any size T. Add line 10, change line 100 and line 150

```

10 INPUT T
100 FOR N = 1 TO T
150 PRINT "AVERAGE ";S/T

```

Run your program for T = 25, for T = 100. To make it easier to read the answers delete line 120. Do you see any difference in the distribution of the sample means? To get a clearer idea we could draw a bar graph, but better yet we will ask the computer to draw it for us. Since the computer is doing the job, we will take 100 samples. Modify your program, change line 30 and 150, and add lines 170 and 180

```

10 INPUT T
15 DIM V(64)
20 FOR M = 1 TO 100
90 LET S = 0
100 FOR N = 1 TO T
110 LET X = INT(100*RND)
130 LET S = S+X
140 NEXT N
150 LET R = INT(S/T)-18
170 LET V(R) = V(R)+1
180 PLOT R, V(R)
200 NEXT M

```

Apple users:

```

5 GT
6 COLOR = 13
110 LET X = INT(100*RND(1))
180 PLOT R-18, V(R)

```

4. Run the program for T = 25 - (Timex-Sinclair users: be sure to operate in the FAST mode!). Run it several times. Can you tell the range where most of the sample averages are?
5. Run the program for T = 100. Do you see what happens when we take a sample four times bigger? Run the program two more times.
- * 6. *This is for just before lunch. Run the program for T=400. Go to lunch and check the output after lunch. Now take a big sample T = 400, Run the program. How does the graph look now? Run the program once more. Would you agree to say that the greater the random sample, the greater is the probability that the sample mean will be close to the real mean?

Mathematics:

Randomness
 Sampling
 Mean
 Sampling

Computers:

INPUT
 FOR-NEXT
 INT
 PLOT

Educated Guessing

1. If we put a number in the boxes such that

$$\boxed{} \times \boxed{} = 56, \text{ (same number in both boxes)}$$

that number is called the square root of 56 and denoted $\sqrt{56}$. Let's guess a number for $\sqrt{56}$ and see how we can get a better guess using the first one.

Since $7 \times 7 = 49$ and $8 \times 8 = 64$ either one would be a good first guess.

Let's say our guess was 7. We know $\sqrt{56}$ is greater than 7 (why?).

So, if we divide $56/7$ we will get a number bigger than 7 (why?).

Now we have two numbers 7 and $56/7$. One is too small and the other too big. If we take the average we will have a better approximation to $\sqrt{56}$.

$$1/2 (7 + 56/7) = 7.5$$

2. Now we can take 7.5 as our guess for $\sqrt{56}$ and repeat the procedure. Use your calculator

$$1/2 (7.5 + 56/7.5) = \boxed{}$$

Now we have an even better approximation. To verify this compute

$$\boxed{} \times \boxed{} = \underline{\hspace{2cm}}$$

3. You can take the number in the box as your guess for $\sqrt{56}$ and repeat the procedure again

$$1/2 (\boxed{} + 56/ \boxed{}) = \boxed{}$$

Compute $\boxed{} \times \boxed{}$ to see how close we are now.

4. We can program the computer to do this procedure for us over and over again.

```
20 INPUT S
30 INPUT R
40 FOR N = 1 TO 12
50 LET R = (R+S/R)/2
60 PRINT R, R*R
70 NEXT N
```

S is the number you want to find the root of. R is your guess. Run the program for the following S: a) 11; b) 7314; c) .87

You can see how fast this algorithm is, even if your first guess is not a very close one.

5. What happens if your guess is very bad? For example, let $S = 56$ and $R = 100$. How many more steps does it take?
- * 6. a) Try $S = 14$ and $R = -1$. What happens? Write down the result.
b) Now try $S = 14$ and $R = 1$. What happens now? Check both answers with a calculator. Are they both "correct"?
- ** 7. Try $S = -14$ and $R = 1$. What happens? Why? Can the product of a number with itself be ever negative?

Mathematics:

Approximation
Iteration

Computers:

INPUT
FOR-TO
NEXT